

シリアルポートを使ったアニメーションの例

Arduinoのアナログ入力にphoto transistor NJL7502Lを接続して、入力光の強さによってProcessingの画面に描いた木の枝角を動かせるプログラムを作ってみよう。

① Arduinoのスケッチは、公開講座5回目で使った簡単なプログラムを使う。photo transistor の接続は添付資料を参照。

② Processing の木を描くプログラムを少し修正して使う。

<http://processing.org/examples/tree.html>

このスケッチを添付のスケッチのようにシリアルポートで送られてきたデータを読み取る部分を追加する。

(注意点) //4*****の行にある番号[5]は各自のArduinoを接続したポート番号に合わせる事。

③ プログラムの動かせ方


- 1) スケッチを実行する前に光センサー回路を組み立てておく。
- 2) まず、Arduinoのスケッチを実行する。
- 3) 続いて、Processingのスケッチを実行する

これで、光センサーに光を当てると枝の角度が変化する。できあがり。

再度実行する時は、

- 4) 一旦、processingのスケッチを停止する（シリアルポートを使っていない状態にする。こうしないと次のArduinoのスケッチが読み込めない！）
- 5) 1) からの手順を繰り返す。

光センサー (Photo Transistor) の接続例



Arduinoのスケッチ

```
// Example 505A
// Serial communication with Processing

//int ledPin=6;
int sensorPin = A0;
int val = 0;

void setup() {
  Serial.begin(9600);
}

void loop() {
  val = analogRead(sensorPin)/4;
  // analogWrite(ledPin, val);
  Serial.write(val);
  delay(100);
}
```

Processingのスケッチ

```
import processing.serial.*; //1*****

Serial port; //2*****
float theta, val; //3*****
```

```

void setup() {
  size(512, 380);
  String arduinoPort = Serial.list()[5]; //4*****
  port = new Serial(this, arduinoPort, 9600); //5*****
}

void draw() {
  background(0,55,255);
  frameRate(30);
  stroke(0,255,0);
  strokeWeight(3);

  if ( port.available() > 0) { //6*****
    val = port.read(); //7*****
    val = map(val, 0, 255, 0, width); //8*****
  } //9*****
  float a = (val / (float) width) * 90f; //10 *****
  theta = radians(a);
  translate(width/2,height);
  line(0,0,0,-120);
  translate(0,-120);
  branch(120);
}

void branch(float h) {
  h *= 0.68;
  if (h > 2) {
    pushMatrix();
    rotate(theta);
    line(0, 0, 0, -h);
    translate(0, -h);
    branch(h);
    popMatrix();

    pushMatrix();
    rotate(-theta);
    line(0, 0, 0, -h);
    translate(0, -h);
    branch(h);
    popMatrix();
  }
}

```

つけたし：この木を描く関数branchは再帰関数となっている。この手法を使っているの
でこんなに短いプログラムでかなり複雑な図形を描くことができる。なお、再帰
(recursive)とは関数の中に自分自身を呼んでいるような、入れ子構造をいう。実際、
上のbranchでは2度branchを呼んでいる。これが左右の枝に対応する。